
PROBLEM: 
A deformable building (an ancient  Greek baptistery in the island of Kos) undergoes 
an internal explosion. This particular building has nothing to do with explosions in 
reality, but has been chosen because a detailed numerical model of the building was 
available from previous seismic studies. 
MESH: 
The fluid (both internal and external to the building) is meshed by 72748 FL34 
tetrahedral elements, the deformable structure by 6894 TETR elements and by 5378 
PRIS elements. The calculation is 3D (TRID) and is declared of the ALE type. An 
absorbing boundary made of 5214 CL3I elements is placed along the outer envelope 
of the fluid mesh to simulate an infinite atmosphere. 
MATERIALS: 
Both the air and the initially solid TNT charge are modelled by the FLUT JWLS 
material (perfect gas and Jones-Wilkins-Lee model), the structure uses a linear elastic 
material (VM23). The absorbing boundary elements have the IMPE ABSI material. 
BOUNDARY CONDITIONS: 
The structure is clamped at the base. FSA and FSR fluid-structure interaction is used 
along the fluid-structure interface as appropriate. 
LOADING: 
Gravity loading in the vertical direction is imposed on the whole model. 
OPTIONS: 
The QUAS STAT option is used to introduce an initial dynamic relaxation phase 
which allows to take into account approximately the static stresses generated in the 
structure by the gravity. This phase goes from t = -20 ms to t = 0. Then follows the 
explosion and the usual transient calculation, until the final time t = 50 ms. 
CALCULATIONS: 

Preliminary calculations involving only the structure are used to estimate the system’s 
frequency under static loading (gravity). Then follows the real calculation as 
described in the previous point. 
RESULTS: 
The initial oscillations are effectively damped out by the quasi-static option. 
POST-TREATMENT 
Animations are produced. 
 
Numerical Solutions 
KOOS21 
The Cast3m mesh generation file reads: 
 
* The Kws Baptistery: 
* tota0: 3D mesh of the structure (qua8, pri6, pyr5, tet4) 
* etota0: 2D mesh of the external face including holes (qua4, tri3) 
* itota0: 2D mesh of the internal face including holes (qua4, tri3) 
* btota0: 2D mesh of the base (qua4, tri3) 
* ttota0: 2D mesh of both faces and base without holes (qua4, tri3) 
* eair0: 3D mesh of the external space (tet4) 
* iair0: 3D mesh of the internal space (tet4) 
* tair0: 3D mesh of both spaces (tet4) 
* Axe x -> South 
* Axe y -> East 
* Axe z -> Up 
*  
ele1 = cub8; 
ele1 = tet4; 
ele2 = tet4; 
opti echo 1 dime 3 elem ele1; 
l0 = 0.25; 
l0 = 0.50; 
dens l0; 
eps0 = 1.d-6; 
* 
*Dimensions 
*Internal radius of the central dome 
r1 = 3.35; 
*External radius of the central dome 
r2 = 3.70; 
*Position of the dome centre (with respect to the cone apex) 
dh0 = 0.35; 
*Height of the cone apex 
h1 = 3.75; 
*Half angle of the external cones 
f0 = 45./2.; 
*Half angle of the internal cones 
g0 = (45.-8.)/2.; 
*External radius of the first ring 
r4 = 3.90; 
*Internal radius of the secong ring 
r5 = 5.00; 
*External radius of the second ring (or external half side) 
r6 = 5.75; 
*Internal radius of the lateral domes 
r7 = 1.50; 
*Height of the lateral domes base 
h7 = h1-r7; 
*Internal radius of the choir dome 
r8 = r7; 
*Height of the choir dome base 
h8 = 2.65; 
h8 = 2.75; 
*Thickness of the choir wall 
dr8 = 0.60; 
*Height of the chapiteau 
dh1 = 0.30; 
*Height of the scale 
he0 = 0.75; 
*Height of the front door 
hd0 = 2.25; 
*Width of the front door 
wd0 = 1.60; 
*Angle of the infill on the big cone 
ai0 = 22.5; 
*Thickness of the infill at the corner 
ti0 = 0.50; 
* 
*Useful quantities 
c45 = cos 45.; 
ca0 = cos ai0; 
sa0 = sin ai0; 
cf0 = cos f0; 
sf0 = sin f0; 
tf0 = sf0/cf0; 
cg0 = cos g0; 
sg0 = sin g0; 
tg0 = sg0/cg0; 
*Internal radius of the first ring 
r3 = (((r1*r1)-(dh0*dh0*cf0*cf0))**0.5 + (dh0*sf0))*cf0; 
*External height of the first ring 
h2 = ((r2*r2)-(r3*r3))**0.5 + dh0 + h1; 
*Intersection sphere r7 / cylindre r5 -> point p2 
k0 = ((2.*r5*r5)-(r7*r7))*c45/r5; 
r0 = ((2.*r5*r5)-(k0*k0))**0.5; 
*External radius of the choir dome 
r9 = (((h1-h8)*(h1-h8))+(r8*r8))**0.5; 
*Half width of the choir window (~square window) 
ww0 = (r6*tg0)+h1-(h8+(((r9*r9)-(r6*r6*(1.-cf0)*(1.-cf0)))**0.5))/2.; 
*Mean radius (where the density will be l0) 
rm0 = (r3+r6)/2.; 
* 
*Definitions of points 
*Axes 
o0 = 0. 0. 0.; 
x0 = 1. 0. 0.; 
y0 = 0. 1. 0.; 
z0 = 0. 0. 1.; 
* 
*Group u: points of the cones at distance rm0 
u0 = rm0 (rm0*tf0) h1; 
u1 = rm0 (rm0*tg0) h1; 
u2 = rm0 0.  h1; 
u3 = rm0 0. (rm0*tg0+h1); 
u4 = rm0 0. (rm0*tf0+h1); 
u5 = rm0 (rm0*tf0*ca0) (rm0*tf0*sa0+h1); 
u6 = (rm0*c45*(1.+(tf0*ca0))) (rm0*c45*(1.-(tf0*ca0))) (rm0*tf0*sa0+h1); 
u7 = (rm0*c45) (rm0*c45) (rm0*tf0+h1); 
u8 = (rm0*c45) (rm0*c45) (rm0*tg0+h1); 
u9 = (rm0*c45) (rm0*c45) h1; 
u10= (rm0*c45*(1.+tg0)) (rm0*c45*(1.-tg0)) h1; 
u11= (rm0*c45*(1.+(tg0*ca0))) (rm0*c45*(1.-(tg0*ca0))) (rm0*tg0*sa0+h1); 
u12= (rm0*c45) (rm0*c45) (ti0*rm0*c45/r6+h1); 
u13= rm0 (ww0*rm0*tf0/r6/tg0) ((1.-(ww0/r6/tg0**2))**0.5*rm0*tf0+h1); 
*Group t: points of dome and top of first ring 
zt0 = h1+dh0; 
t0 = 0. 0. zt0; 
t1 = (r3*r1/r2) 0. ((h2-zt0)*r1/r2 + zt0); 
t2 = 0. 0. (zt0+r1); 
t3 = (r3*c45*r1/r2) (r3*c45*r1/r2) ((h2-zt0)*r1/r2 + zt0); 

t4 = r3 0. h2; 
t6 = (r3*c45) (r3*c45) h2; 
t7 = r4 0. h2; 
t8 = (r4*c45) (r4*c45) h2; 
*Group a: points of the internal first ring (r3) 
a0 = 0. 0. h1; 
a1 = (r3*cg0) (r3*sg0) h1; 
a4 = r3 0. (r3*tf0+h1); 
a7 = (r3*c45) (r3*c45) (r3*tf0+h1); 
*Group b: points of the external first ring (r4) 
b4 = r4 0. (r4*tf0+h1); 
b7 = (r4*c45) (r4*c45) (r4*tf0+h1); 
*Group c: points of the internal second ring (r5) 
c1 = (r5*cg0) (r5*sg0) h1; 
c9 = (r5*c45) (r5*c45) h1; 
c10= (r5*c45*(cg0+sg0)) (r5*c45*(cg0-sg0)) h1; 
*Group d: points of the external second ring (r6) 
d0 = (r6*cf0) (r6*sf0) h1; 
d1 = (r6*cf0) (r6*cf0*tg0) h1; 
d2 = (r6*cf0) 0. h1; 
d9 = (r6*c45) (r6*c45) h1; 
*Group e: points of the external side (r6) 
e0 = r6 (r6*tf0) h1; 
e1 = r6 (r6*tg0) h1; 
e9 = r6 r6 h1; 
e10= r6 (r6*(cg0-sg0)/(cg0+sg0)) h1; 
e12= r6 r6 (ti0+h1); 
e13= r6 ww0 (((r6*r6*tg0*tg0)-(ww0*ww0))**0.5+h1); 
*Group q: points at level (h1-dh1) 
zq0 = h1-dh1; 
q0 = 0. 0. zq0; 
q1 = (r5*cg0) (r5*sg0) zq0; 
q2 = (r5*c45*(cg0+sg0)) (r5*c45*(cg0-sg0)) zq0; 
q7 = r6 (r6*tg0) zq0; 
q8 = (r6*cf0) (r6*cf0*tg0) zq0; 
*Group p: points at level (h1-r7) 
zp0 = h1-r7; 
p0 = 0. 0. zp0; 
p1 = (r5*cg0) (r5*sg0) zp0; 
p2 = ((k0+r0)/2.) ((k0-r0)/2.) zp0; 
p3 = (r5*c45) (r5*c45) zp0; 
p4 = ((r5+r7)*c45) ((r5+r7)*c45) zp0; 
p5 = r6 r6 zp0; 
p7 = r6 (r6*tg0) zp0; 
*Group i,j,k relative to the choir 
i0 = 0. 0. h8; 
i1 = (r6*cf0) 0. h8; 
i2 = (r6*cf0) 0. (h8+r8); 
i3 = ((r6*cf0)+(r8*r8/r9)) 0. (((h1-h8)*r8/r9)+h8); 
i5 = ((r6*cf0)+r8) 0. h8; 
i6 = (r6*cf0) r8 h8; 
i7 = r6 (r6*tg0) h8; 
j1 = r6 0. h8; 
j3 = ((r6*cf0)+r8) 0. h1; 
j6 = (r6*cf0) (r6*cf0*tg0) h8; 
j7 = r6 0.  (((r9*r9)-(r6*r6*(1.-cf0)*(1.-cf0)))**0.5+h8); 
j78= r6 ww0 (((r9*r9)-(r6*r6*(1.-cf0)*(1.-cf0))-(ww0*ww0))**0.5+h8); 
j8 = r6 (((r8*r8)-(r6*r6*(1.-cf0)*(1.-cf0)))**0.5) h1; 
k3 = ((r6*cf0)+r8+dr8) 0. h1; 
k4 = ((r6*cf0)+r8+dr8) (r6*tg0/2.) h1; 
k5 = ((r6*cf0)+r8+dr8) 0. h8; 
*Group w relative to the opening 
*Front door 
w0 = r6 0. 0.; 
w1 = r6 0. he0; 
w2 = r6 (wd0/2.) he0; 
w3 = r6 (wd0/2.) (hd0+he0); 
w4 = r6 0. (hd0+he0); 
w7 = r6 (r6*tg0) 0.; 
*Number of elements generated between two homeomorphic lines/surfaces 
*Within the cone thickness 
nn0 = 1+(enti (tf0-tg0*rm0/l0)); 
*Within the first ring 
nn1 = 1+(enti (r4-r3/l0)); 
*Between the first and second ring 
nn2 = 1+(enti (r5-r4/l0)); 
*Within the second ring 
nn3 = 1+(enti (r6-r5/l0)); 
*Within the choir wall thicness 
nn4 = 1+(enti (r6/cf0-r6/l0)); 
*within the choir dome thickness 
nn5 = 1+(enti (r9-r8/l0)); 
* 
*Reference lines 
u0u1 = d u0 u1 dini l0 dfin l0; 
u1u2 = d u1 u2 dini l0 dfin l0; 
u2u3 = d u2 u3 dini l0 dfin l0; 
u0u5 = cerc u0 u2 u5 dini l0 dfin l0; 
u5u13= cerc u5 u2 u13 dini l0 dfin l0; 
u13u4= cerc u13 u2 u4 dini l0 dfin l0; 
u5u4 = u5u13 et u13u4; 
u0u4 = u0u5 et u5u4; 
u1u13= (u0u5 et u5u13) homo (tg0/tf0) u2; 
u13u3= u13u4 homo (tg0/tf0) u2; 
u1u3 = u0u4 homo (tg0/tf0) u2; 
u3u4 = d u3 u4 dini l0 dfin l0; 
u5u6 = d u5 u6 dini l0 dfin l0; 
u6u7 = cerc u6 u9 u7 dini l0 dfin l0; 
u11u8= u6u7 homo (tg0/tf0) u9; 
u8u7 = d u8 u7 dini l0 dfin l0; 
u12u8= d u12 u8 dini l0 dfin l0; 
u9u12= d u9 u12 dini l0 dfin l0; 
u10u9= d u10 u9 dini l0 dfin l0; 
u11u12=d u11 u12 dini l0 dfin l0; 
u0u10= d u0 u10 dini l0 dfin l0; 
u6u11= d u6 u11 dini l0 dfin l0; 
u0u6 = cerc u0 u9 u6 dini l0 dfin l0; 
u0u7 = u0u6 et u6u7; 
u10u11=u0u6 homo (tg0/tf0) u9; 
u10u8= u0u7 homo (tg0/tf0) u9; 
u1u10= d u1 u10 dini l0 dfin l0; 
*Reference surfaces 
elim eps0 (u0u1 et u1u2 et u2u3 et u0u5 et u5u4 et u0u4 et   
           u1u13 et u1u3 et u3u4 et u5u6 et u6u7 et u11u8 et 
           u8u7 et u12u8 et u9u12 et u10u9 et u11u12 et 
           u0u10 et u0u6 et u10u11 et u10u8 et u1u10); 
sur1 = (u1u2 et u2u3 et u1u3) surf plan coul bleu; 

sur2 = u0u4 regl nn0 u1u3 coul roug; 
opti elem ele2; 
sur3 =((u5u6 et u0u5 et u0u6) proj coni a0 cyli a0 t0 u2) 
                              surf cylindrique a0 t0 coul vert; 
opti elem ele1; 
sur4 = u10u11 regl nn0 u0u6 coul turq; 
sur5 = u11u8 regl nn0 u6u7 coul rose; 
sur6 = (u10u11 et u11u12 et u9u12 et u10u9) surf plan coul jaun; 
sur7 = (u11u8 et u11u12 et u12u8) surf plan coul blan; 
*trak (sur1 et sur2 et sur3 et sur4 et sur5 et sur6 et sur7); 
* 
*Construction of the conjunction between the dome and the cones (named truc0 
here) 
opti elem ele2; 
a1a10= u1u10 proj coni a0 cyli a0 t0 a4; 
a1a3 = u1u3 proj coni a0 cyli a0 t0 a4; 
a3a4 = u3u4 proj coni a0 cyli a0 t0 a4; 
a4a7 = cerc a4 (0. 0. (r3*tf0+h1)) a7 dini l0 dfin l0; 
a8a7 = u8u7 proj coni a0 cyli a0 t0 a4; 
a10a8= u10u8 proj coni a0 cyli a0 t0 a4; 
tru1 = a1a10 et a1a3 et a3a4 et a4a7 et a8a7 et a10a8; 
elim eps0 tru1; 
tru1 = tru1 surf cylindrique a0 t0; 
* 
a4t1 = cerc a4 t0 t1 dini l0 dfin l0; 
t1t3 = cerc t1 (0. 0. ((h2-zt0)*r1/r2 + zt0)) t3 dini l0 dfin l0; 
a7t3 = cerc a7 t0 t3 dini l0 dfin l0; 
tru2 = a4a7 et a4t1 et t1t3 et a7t3; 
elim eps0 tru2; 
tru2 = tru2 surf spherique a0; 
* 
t1t4 = d t1 t4 dini l0 dfin l0; 
*t4t6 = cerc t4 (0. 0. h2) t6 dini l0 dfin l0; 
t4t6 = t1t3 homo (r2/r1) t0; 
t3t6 = d t3 t6 dini l0 dfin l0; 
tru3 = t1t3 et t1t4 et t4t6 et t3t6; 
elim eps0 tru3; 
tru3 = tru3 surf conique t0 a0; 
tru3 = t1t3 regl t4t6 dini l0 dfin l0; 
* 
t4t7 = d t4 t7 dini l0 dfin l0; 
t7t8 = cerc t7 (0. 0. h2) t8 dini l0 dfin l0; 
t6t8 = d t6 t8 dini l0 dfin l0; 
tru4 = t4t6 et t4t7 et t7t8 et t6t8; 
elim eps0 tru4; 
tru4 = tru4 surf plan; 
* 
b7t8 = d b7 t8 dini l0 dfin l0; 
b4t7 = d b4 t7 dini l0 dfin l0; 
b6b7 = u6u7 proj coni a0 cyli a0 t0 b4; 
b5b6 = u5u6 proj coni a0 cyli a0 t0 b4; 
b5b4 = u5u4 proj coni a0 cyli a0 t0 b4; 
tru5 = t7t8 et b7t8 et b6b7 et b5b6 et b5b4 et b4t7; 
elim eps0 tru5; 
tru5 = tru5 surf cylindrique a0 t0; 
* 
b3b4 = u3u4 proj coni a0 cyli a0 t0 b4; 
tru6 = a3a4 regl nn1 b3b4; 
a4b4 = tru6 cote 2; 
elim eps0 (t1t4 et t4t7 et b4t7 et a4b4 et a4t1); 
tru6 = ((t1t4 et t4t7 et b4t7 et a4b4 et a4t1) surf plan) et tru6; 
* 
b8b7 = u8u7 proj coni a0 cyli a0 t0 b4; 
tru7 = b8b7 regl nn1 a8a7; 
a7b7 = tru7 cote 2; 
elim eps0 (t3t6 et t6t8 et b7t8 et a7b7 et a7t3); 
tru7 = ((t3t6 et t6t8 et b7t8 et a7b7 et a7t3) surf plan) et tru7; 
* 
b1b3 = u1u3 proj coni a0 cyli a0 t0 b4; 
tru8 = a1a3 regl nn1 b1b3; 
* 
b10b8= u10u8 proj coni a0 cyli a0 t0 b4; 
tru9 = a10a8 regl nn1 b10b8; 
* 
a1b1 = tru8 cote 4; 
a10b10=tru9 cote 4; 
b0b10= u0u10 proj coni a0 cyli a0 t0 b4; 
b0b1 = u0u1 proj coni a0 cyli a0 t0 b4; 
tru10= a1b1 et b0b1 et b0b10 et a10b10 et a1a10; 
elim eps0 tru10; 
tru10= tru10 surf plan; 
* 
tru11 =((sur2 et sur4 et sur5) chan ele2) proj coni a0 cyli a0 t0 b4; 
tru12 = sur3 proj coni a0 cyli a0 t0 b4; 
truc0 = tru1 et tru2 et tru3 et tru4 et tru5 et tru6 et  
        tru7 et tru8 et tru9 et tru10 et tru11 et tru12; 
elim eps0 truc0; 
truc0 = truc0 volu coul rose; 
*trak truc0; 
itruc0 = tru1 et tru2 et tru8 et tru9 coul rose; 
etruc0 = tru4 et tru5 coul rose; 
*trak (itruc0 et etruc0); 
* 
*Construction of the central dome (named calo0 here); 
opti elem ele1; 
t1t2 = cerc t1 t0 t2 dini l0 dfin l0; 
t3t2 = cerc t3 t0 t2 dini l0 dfin l0; 
cal1 = t1t3 et t3t2 et t1t2; 
elim eps0 cal1; 
cal1 = cal1 surf spherique t0; 
cal2 = cal1 homo (r2/r1) t0; 
calo0 = cal1 volu cal2 dini l0 dfin l0 coul bleu; 
*trak calo0; 
icalo0 = cal1 coul bleu; 
ecalo0 = cal2 coul bleu; 
*trak (icalo0 et ecalo0); 
* 
*Contruction of the small cone (named scon0 here) 
sco1 = (sur4 et sur5) proj coni a0 cyli a0 t0 b4; 
sco2 = sco1 homo (r5/r4) a0; 
sco3 = sco2 homo (r6/r5) a0; 
sco4 = (sur6 et sur7) proj coni a0 cyli a0 t0 c1; 
sco5 = sco4 homo (r6/r5) a0; 
scon0 = (sco1 volu nn2 sco2  
              volu nn3 sco3) et  
        (sco4 volu nn3 sco5) coul bleu; 
elim eps0 scon0; 
*trak scon0; 
sco6 = regl nn2 (u10u8 proj coni a0 cyli a0 t0 b4) 
                (u10u8 proj coni a0 cyli a0 t0 c1); 
sco7 = regl nn2 (u6u7 proj coni a0 cyli a0 t0 b4) 
                (u6u7 proj coni a0 cyli a0 t0 c1) 
       regl nn3 (u6u7 proj coni a0 cyli a0 t0 d0); 
sco8 = (sur5 et sur7) proj coni a0 cyli a0 t0 d0; 
iscon0 = sco4 et sco6 coul bleu; 
escon0 = sco7 et sco8 coul bleu; 
elim eps0 (iscon0 et escon0); 
*trak (iscon0 et escon0); 
* 
*Contruction of the big cone (named bcon0 here) 
bco1 = sur2 proj coni a0 cyli a0 t0 b4; 
bco2 = bco1 homo (r5/r4) a0; 
bco3 = bco2 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0); 
bco4 = bco3 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
bco5 = sur1 proj coni a0 cyli a0 t0 c1; 
bco6 = bco5 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0); 
bco7 = bco5 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
bcon0 = (bco1 volu nn2 bco2 
              volu nn3 bco3 
              volu nn4 bco4) et 
        (bco5 volu nn3 bco6 
              volu nn4 bco7) coul roug; 
elim eps0 bcon0; 
*trak bcon0; 
bco8 = regl nn2 (u1u3 proj coni a0 cyli a0 t0 b4) 
                (u1u3 proj coni a0 cyli a0 t0 c1); 
bco9 = regl nn2 (u5u4 proj coni a0 cyli a0 t0 b4) 
                (u5u4 proj coni a0 cyli a0 t0 c1) 
       regl nn3 (u5u4 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0)) 
       regl nn4 (u5u4 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0)); 
ibcon0 = bco5 et bco8 coul roug; 
ebcon0 = bco4 et bco7 et bco9 coul roug; 
elim eps0 (ibcon0 et ebcon0); 
*trak (ibcon0 et ebcon0); 

* 
*Construction of the filling (named fill0 here) 
fil1 = sur3 proj coni a0 cyli a0 t0 b4; 
fil2 = fil1 homo (r5/r4) a0; 
fil3 = fil2 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0); 
fil4 = fil3 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
d0d6 = u0u6 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0); 
elim eps0 (fil3 et d0d6); 
depl d0d6 proj coni a0 cyli a0 t0 d0; 
fil5 = sur4 proj coni a0 cyli a0 t0 d0; 
fil6 = fil5 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
opti elem ele2; 
si (sur6 elem type dime neg 1); 
  sur6 = (sur6 elem tri3) et (sur6 elem qua4 'NOVERIF' chan tri3); 
fins; 
fil7 = sur6 proj coni a0 cyli a0 t0 d0; 
e10e11 = u10u11 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
e9e12 = u9u12 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
e10e9 = d e10 e9 dini l0 dfin l0; 
e11e12= d (e10e11 poin final) e12 dini l0 dfin l0; 
fil8 = e10e9 et e10e11 et e11e12 et e9e12; 
elim eps0 fil8; 
fil8 = fil8 surf plan; 
d10d11= u10u11 proj coni a0 cyli a0 t0 d0; 
e10e11= u10u11 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
fil9 = d10d11 regl nn4 e10e11; 
d9d12= u9u12 proj coni a0 cyli a0 t0 d0; 
e9e12= u9u12 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
fil10 = d9d12 regl e9e12 dini l0 dfin l0; 
d12e12= fil10 cote 2; 
d11e11= fil9 cote 2; 
d11d12= u11u12 proj coni a0 cyli a0 t0 d0; 
fil11 = d11e11 et d11d12 et d12e12 et e11e12; 
elim eps0 fil11; 
fil11 = fil11 surf plan; 
d9e9 = fil10 cote 4; 
d10e10 = fil9 cote 4; 
d10d9 = u10u9 proj coni a0 cyli a0 t0 d0; 
fil12 = d9e9 et d10e10 et d10d9 et e10e9; 
elim eps0 fil12; 
fil12 = fil12 surf plan; 
fill0 = fil7 et fil8 et fil9 et fil10 et fil11 et fil12; 
elim eps0 fill0; 
fill0 = fill0 volu; 
opti elem ele1; 
fill0 = fill0 et (fil1 volu nn2 fil2  
                       volu nn3 fil3 
                       volu nn4 fil4) et 
                 (fil5 volu nn4 fil6) coul vert; 
elim eps0 fill0; 
*trak fill0; 
c5c6 = b5b6 homo (r5/r4) a0; 
d5d6 = c5c6 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0); 
d6 = u6 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0); 
elim eps0 (d5d6 et d6); 
depl d6 proj coni a0 cyli a0 t0 d0; 
e5e6 = d5d6 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
fil13 = b5b6 regl nn2 c5c6 regl nn3 d5d6 regl nn4 e5e6; 
fil14 = regl (u6u11 proj coni a0 cyli a0 t0 d0) nn4  
             (u6u11 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0)); 
efill0 = fil4 et fil6 et fil8 et fil11 et fil13 et fil14 coul vert; 
elim eps0 efill0; 
*trak efill0; 
* 
*Construction of the chapiteau (named chap0 here) 
c1c10= (b0b10 et b0b1) homo (r5/r4) a0; 
cha1 = (b0b10 et b0b1) regl nn2 c1c10; 
chap0 = tru10 et cha1; 
elim eps0 chap0; 
chap0 = chap0 volu tran (q0 moin a0) dini l0 dfin l0 coul vert; 
*trak chap0; 
cha2 = cha1 plus (q0 moin a0); 
cha3 = (c1 droi nn2 a1b1) et a1a10 et (c10 droi nn2 a10b10) 
       tran (q0 moin a0) dini l0 dfin l0; 
ichap0 = cha2 et cha3 coul vert; 
elim eps0 ichap0; 
*trak ichap0; 
* 
*Construction of the column (named colo0 here) 
col1 = tru10 moin a0; 
colo0 = col1 volu tran q0 dini l0 dfin l0 coul roug; 
*trak colo0; 
icolo0 = cont col1 tran q0 dini l0 dfin l0 coul roug; 
bcolo0 = col1 coul roug; 
elim eps0 (icolo0 et bcolo0); 
*trak (bcolo0 et icolo0); 
* 
*Construction of the wall under the big cone (named bwal0 here) 
c1c2 = u1u2 proj coni a0 cyli a0 t0 c1; 
d1d2 = u1u2 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0); 
e1e2 = u1u2 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
bwa1 = c1c2 regl nn3 d1d2 regl nn4 e1e2; 
bwal0 = bwa1 volu tran (q0 moin a0) dini l0 dfin l0 
             volu tran (i0 moin q0) dini l0 dfin l0 
             volu tran (p0 moin i0) dini l0 dfin l0 
             volu tran (o0 moin p0) dini l0 dfin l0 coul turq; 
*trak bwal0; 
ibwal0 = c1c2 tran (q0 moin a0) dini l0 dfin l0 
              tran (i0 moin q0) dini l0 dfin l0 
              tran (p0 moin i0) dini l0 dfin l0 
              tran (o0 moin p0) dini l0 dfin l0 coul turq; 
ebwal0 = e1e2 tran (q0 moin a0) dini l0 dfin l0 
              tran (i0 moin q0) dini l0 dfin l0 
              tran (p0 moin i0) dini l0 dfin l0 
              tran (o0 moin p0) dini l0 dfin l0 coul turq; 
bbwal0 = bwa1 plus o0 moin a0 coul turq; 
elim eps0 (ibwal0 et ebwal0 et bbwal0); 
*trak (ibwal0 et ebwal0 et bbwal0); 
* 
*Construction of the wall under the small cone+chapiteau (named swal0 here) 
opti elem ele2; 
d1d10=(u0u10 proj coni a0 cyli a0 t0 d0) et 
      (u0u1 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0)); 
elim eps0 d1d10; 
e1e10= c1c10 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
swa1 = c1c10 regl nn3 d1d10 regl nn4 e1e10; 
c10c9 = u10u9 proj coni a0 cyli a0 t0 c1; 
d10d9 = u10u9 proj coni a0 cyli a0 t0 d0; 
swa2 = fil12 et (c10c9 regl nn3 d10d9); 
swa3 = c1c10 tran (q0 moin a0) dini l0 dfin l0; 
c1d1 = c1 d nn3 d1; 
swa4 = c1d1 tran (q0 moin a0) dini l0 dfin l0 
            tran (i0 moin q0) dini l0 dfin l0 
            tran (p0 moin i0) dini l0 dfin l0; 
d1e1 = d1 d nn4 e1; 
swa5 = d1e1 tran (q0 moin a0) dini l0 dfin l0 
            tran (i0 moin q0) dini l0 dfin l0 
            tran (p0 moin i0) dini l0 dfin l0; 
e1e9 = e1e10 et e10e9; 
elim eps0 e1e9; 
swa6 = e1e9 tran (q0 moin a0) dini l0 dfin l0  
            tran (i0 moin q0) dini l0 dfin l0 
            tran (p0 moin i0) dini l0 dfin l0; 
c9e9 = (d nn3 c9 d9) et d9e9; 
p5e9 = swa6 cote 2; 
p4p5 = d p4 p5 dini l0 dfin l0; 
c9p4 = cerc c9 p3 p4 dini l0 dfin l0; 
swa7 = c9e9 et p5e9 et p4p5 et c9p4; 
elim eps0 swa7; 
swa7 = swa7 surf plan; 
q2c10= d q2 c10 dini l0 dfin l0; 
q1q2 = c1c10 plus (q0 moin a0); 
q1p1 = d q1 (q1 plus i0 moin q0) dini l0 dfin l0 d p1 dini l0 dfin l0; 
p1p2 = d p1 p2 dini l0 dfin l0; 
p2c9 = cerc p2 p3 c9 dini l0 dfin l0; 
p2p3 = (p2c9 proj cyli a0 plan p2 p3 p4) 
             proj coni p0 cyli p0 a0 p3; 
p2c9 = (p2p3 proj cyli a0 sphe p3 p2) syme plan p0 p2 p3; 
swa8 = c10c9 et q2c10 et q1q2 et q1p1 et p1p2 et p2c9; 
elim eps0 swa8; 
swa8 = swa8 surf cylindrique a0 t0; 
p2p4 = cerc p2 p3 p4 dini l0 dfin l0;; 
p7p5 = e1e9 plus (p0 moin a0); 
p1p7 = (c1d1 et d1e1) plus (p0 moin a0); 

swa9 = p1p2 et p2p4 et p4p5 et p7p5 et p1p7; 
elim eps0 swa9; 
swa9 = swa9 surf plan; 
swa0 = c9p4 et p2c9 et p2p4; 
elim eps0 swa0; 
swa0 = swa0 surf spherique p3; 
swal0 = swa1 et swa2 et swa3 et swa4 et swa5 et 
        swa6 et swa7 et swa8 et swa9 et swa0; 
elim eps0 swal0; 
swal0 = swal0 volu; 
opti elem ele1; 
swal0 = swal0 et 
       (swa9 volu tran (o0 moin p0) dini l0 dfin l0) coul rose; 
elim eps0 swal0; 
*trak swal0; 
iswal0 =((p1p2 et p2p4) tran (o0 moin p0) dini l0 dfin l0) 
         et swa8 et swa0 coul rose; 
eswal0 =( p7p5          tran (o0 moin p0) dini l0 dfin l0) 
         et swa6 coul rose; 
bswal0 = swa9 plus o0 moin p0 coul rose; 
elim eps0 (iswal0 et eswal0 et bswal0); 
*trak (iswal0 et eswal0 et bswal0); 
* 
*Construction of the alternative of bcon0 (named bcon1 here) 
e1e13= u1u13 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
j8j78= cerc j8 j1 j78 dini l0 dfin l0; 
e1j8 = d e1 j8 dini l0 dfin l0; 
e13j78 = d e13 j78 dini l0 dfin l0; 
bco11 = j8j78 et e13j78 et e1e13 et e1j8; 
elim eps0 bco11; 
dens 1.d-6; 
bco11 = bco11 surf plan; 
bco12 = bco11 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0); 
bcon1 = (bco1 volu nn2 bco2 
              volu nn3 bco3 
              volu nn4 bco4) et 
        (bco11 volu nn4 bco12) coul jaun; 
*trak bcon1; 
bco18 = bco8 regl nn3 
       (u1u3 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0)); 
d13j78 = e13j78 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0); 
d13d3 = u13u3 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0); 
e13e3 = u13u3 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
bco19 = (e13j78 et e13e3) regl nn4 (d13j78 et d13d3); 
ibcon1 = bco18 et bco19 et bco12 coul jaun; 
ebcon1 = bco4 et bco9 et bco11 coul jaun; 
elim eps0 (ibcon1 et ebcon1); 
*Window 
e2e3 = u2u3 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
j78j7= j78 cerc j1 j7 dini l0 dfin l0; 
e3 = u3 proj coni a0 plan e0 (e0 plus y0) (e0 plus z0); 
j7e3 = j7 droi e3 dini l0 dfin l0; 
void1 = e13j78 et j78j7 et j7e3 et e13e3; 
elim eps0 void1; 
void1 = void1 surf plan coul blan; 
elim eps0 (ibcon1 et ebcon1 et void1); 
*trak (ibcon1 et ebcon1 et void1); 
* 
*Construction of the choir (named choi0 here) 
j8j7 = j8j78 et j78j7; 
j2j4 = j8j7 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0); 
cho1 = j8j7 regl nn4 j2j4; 
cho2 = cho1 proj coni i1 sphe i1 i2; 
j3j7 = cerc j3 i1 j7 dini l0 dfin l0; 
j3j8 = cerc j3 d2 j8 dini l0 dfin l0; 
cho3 = j8j7 et j3j7 et j3j8; 
elim eps0 cho3; 
cho3 = cho3 surf spherique i1; 
cho4 = cho3 proj coni i1 sphe i1 i2; 
choi0 = (cho1 volu nn5 cho2) et  
        (cho3 volu nn5 cho4) coul bleu; 
elim eps0 choi0; 
*trak choi0; 
j78j7bi = j78j7 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0); 
cho5 = j78j7 regl nn4 j78j7bi; 
j2j4bi = j2j4 proj coni i1 sphe i1 i2; 
cho6 = j2j4 regl nn5 j2j4bi; 
ichoi0 = cho2 et cho4 et cho5 et cho6 coul bleu; 
echoi0 = cho3 coul bleu; 
elim eps0 (ichoi0 et echoi0); 
*trak echoi0; 
* 
*Construction of the first alternative of bwal0 (named bwal1 here) 
opti elem ele2; 
d1j4 = e1j8 proj coni a0 plan d0 (d0 plus y0) (d0 plus z0); 
bwa11 = e1j8 regl nn4 d1j4; 
j8j4 = bwa11 cote 2; 
i8i4 = j8j4 proj coni i1 sphe i1 i2; 
bwa12 = j8j4 regl nn5 i8i4; 
i3i8 = j3j8 homo i1 (r8/r9); 
bwa13 = j3j8 regl nn5 i3i8; 
j3k3 = d j3 k3 dini l0 dfin l0; 
k3k4 = d k3 k4 dini l0 dfin l0; 
e1k4 = d e1 k4 dini l0 dfin l0; 
bwa14 = e1j8 et j3j8 et j3k3 et k3k4 et e1k4; 
elim eps0 bwa14; 
bwa14 = bwa14 surf plan; 
j4i4 = bwa12 cote 2; 
j4i6 = cerc j4i4 i1 i6 dini l0 dfin l0; 
i6j6 = d i6 j6 dini l0 dfin l0; 
d1j6 = d d1 q8 dini l0 dfin l0 d j6 dini l0 dfin l0; 
bwa15= d1j4 et j4i6 et i6j6 et d1j6; 
elim eps0 bwa15; 
bwa15= bwa15 surf plan; 
d1e1 = bwa11 cote 4; 
bwa16= d1e1 tran (q0 moin a0) dini l0 dfin l0 
            tran (i0 moin q0) dini l0 dfin l0; 
e1k3 = e1k4 et k3k4; 
elim eps0 e1k3; 
bwa17= e1k3 tran (q0 moin a0) dini l0 dfin l0 
            tran (i0 moin q0) dini l0 dfin l0; 
k3k5 = d1j6 plus k3 moin d1; 
k5i5 = d k5 i5 dini l0 dfin l0; 
i3i5 = cerc i3 i1 i5 dini l0 dfin l0; 
i3j3 = bwa13 cote 4; 
bwa18 = j3k3 et k3k5 et k5i5 et i3i5 et i3j3; 
elim eps0 bwa18; 
bwa18 = bwa18 surf plan; 
j6k5 = (d1e1 et e1k3) plus i0 moin a0; 
i5i6 = cerc i5 i1 i6 dini l0 dfin l0; 
bwa19 = i6j6 et j6k5 et k5i5 et i5i6; 
elim eps0 bwa19; 
bwa19 = bwa19 surf plan; 
i8i6 = cerc i8i4 i1 i6 dini l0 dfin l0; 
bwa10 = i5i6 et i3i5 et i3i8 et i8i6; 
elim eps0 bwa10; 
bwa10 = bwa10 surf spherique i1; 
bwal1 = bwa11 et bwa12 et bwa13 et bwa14 et bwa15 et 
        bwa16 et bwa17 et bwa18 et bwa19 et bwa10; 
elim eps0 bwal1; 
bwal1 = bwal1 volu; 
opti elem ele1; 
bwal1 = bwal1 et  
       (bwa19 volu tran (p0 moin i0) dini l0 dfin l0 
              volu tran (o0 moin p0) dini l0 dfin l0) coul turq; 
elim eps0 bwal1; 
*trak bwal1; 
e1k3bi= e1k3 plus i0 moin a0;  
swa14 =((c1d1 plus p0 moin a0) tran (o0 moin p0) dini l0 dfin l0) 
       et swa4 coul rose; 
ibwal1=((i5i6 et i6j6) tran (p0 moin i0) dini l0 dfin l0 
                       tran (o0 moin p0) dini l0 dfin l0) et 
         bwa10 et bwa15 coul turq et swa14; 
ebwal1= (e1k3bi tran (p0 moin i0) dini l0 dfin l0 
                tran (o0 moin p0) dini l0 dfin l0) et 
         bwa14 et bwa17 coul turq; 
bbwal1=  bwa19 plus o0 moin i0 coul turq; 
elim eps0 (ibwal1 et ebwal1 et bbwal1); 
*trak (ibwal1 et ebwal1 et bbwal1); 
* 
*Construction of the entrance under the big cone (named bwal2 here) 
por1 = w4 droi w3 dini l0 dfin l0 droi w2 dini l0 dfin l0 
          droi w1 dini l0 dfin l0; 
e1e2bi = (e1e2 plus o0 moin a0) inve; 

bwa21 = e1e2 droi por1 dini l0 dfin l0 droi e1e2bi dini l0 dfin l0 
             droi p7   dini l0 dfin l0 droi i7     dini l0 dfin l0 
             droi q7   dini l0 dfin l0 droi        dini l0 dfin l0; 
bwa21 = bwa21 surf plan; 
coo1 coo2 coo3 = coor bwa21; 
de12 = (nomc ux ((cf0-1.)*coo1)) + (nomc uy ((cf0-1.)*coo2)); 
bwa22 = bwa21 plus de12; 
ra13 = ((coo2/r6)**2+1.)**(-0.5)*r5; 
de13 = (nomc ux (ra13-coo1)) + (nomc uy (ra13*coo2/r6-coo2)); 
bwa23 = bwa21 plus de13; 
bwal2 = bwa21 volu nn4 bwa22 volu nn3 bwa23 coul turq; 
*trak bwal2; 
por2 = por1 plus de12; 
por3 = por1 plus de13; 
bwa24 = por1 regl nn4 por2 regl nn3 por3; 
ibwal2 = bwa23 et bwa24 coul turq; 
ebwal2 = bwa21 coul turq; 
bbwal2 = bbwal0; 
elim eps0 (ibwal2 et ebwal2 et bbwal2); 
*Door 
void2 = por1 droi dini l0 dfin l0; 
void2 = void2 surf plan coul blan; 
*trak (ibwal2 et ebwal2 et bbwal2 et void2);  
* 
mesh0 = truc0 et calo0 et scon0 et bcon0 et fill0 et  
        chap0 et colo0 et bwal0 et swal0;         
elim eps0 mesh0; 
*trak mesh0; 
imesh0 = itruc0 et icalo0 et iscon0 et ibcon0 et  
         ichap0 et icolo0 et ibwal0 et iswal0;          
emesh0 = etruc0 et ecalo0 et escon0 et ebcon0 et  
         efill0 et ebwal0 et eswal0; 
bmesh0 = bcolo0 et bbwal0 et bswal0; 
elim eps0 (imesh0 et emesh0 et bmesh0); 
*trak (imesh0 et emesh0 et bmesh0); 
* 
mesh1 = truc0 et calo0 et scon0 et bcon1 et fill0 et  
        chap0 et colo0 et bwal1 et swal0 et choi0; 
elim eps0 mesh1; 
*trak mesh1; 
imesh1 = itruc0 et icalo0 et iscon0 et ibcon1 et ichap0 et  
         icolo0 et ibwal1 et iswal0 et ichoi0 et void1; 
emesh1 = etruc0 et ecalo0 et escon0 et ebcon1 et   
         efill0 et ebwal1 et eswal0 et echoi0 et void1; 
bmesh1 = bcolo0 et bbwal1 et bswal0; 
elim eps0 (imesh1 et emesh1 et bmesh1); 
*trak (imesh1 et emesh1 et bmesh1); 
* 
mesh2 = truc0 et calo0 et scon0 et bcon0 et fill0 et  
        chap0 et colo0 et swal0 et bwal2; 
elim eps0 mesh2; 
*trak mesh2; 
imesh2 = itruc0 et icalo0 et iscon0 et ibcon0 et 
         ichap0 et icolo0 et iswal0 et ibwal2 et void2; 
emesh2 = etruc0 et ecalo0 et escon0 et ebcon0 et  
         efill0 et eswal0 et ebwal2 et void2; 
bmesh2 = bcolo0 et bswal0 et bbwal2; 
elim eps0 (imesh2 et emesh2 et bmesh2); 
*trak (imesh2 et emesh2 et bmesh2); 
* 
demi0 = mesh1 et (mesh0 syme plan o0 (x0 plus y0) z0) 
              et (mesh2 syme plan o0 y0 z0) 
              et (mesh0 tour 90. o0 z0); 
elim eps0 demi0; 
*trak (enve demi0); 
idemi0 = imesh1 et (imesh0 syme plan o0 (x0 plus y0) z0) 
                et (imesh2 syme plan o0 y0 z0) 
                et (imesh0 tour 90. o0 z0); 
edemi0 = emesh1 et (emesh0 syme plan o0 (x0 plus y0) z0) 
                et (emesh2 syme plan o0 y0 z0) 
                et (emesh0 tour 90. o0 z0); 
bdemi0 = bmesh1 et (bmesh0 syme plan o0 (x0 plus y0) z0) 
                et (bmesh2 syme plan o0 y0 z0) 
                et (bmesh0 tour 90. o0 z0); 
elim eps0 (idemi0 et edemi0 et bdemi0); 
*trak (idemi0 et edemi0 et bdemi0); 
tota0 = demi0 et (demi0 syme plan o0 x0 z0); 
elim eps0 tota0; 
*trak (enve tota0); 
itota0 = idemi0 et (idemi0 syme plan o0 x0 z0); 
etota0 = edemi0 et (edemi0 syme plan o0 x0 z0); 
btota0 = bdemi0 et (bdemi0 syme plan o0 x0 z0); 
elim eps0 (itota0 et etota0 et btota0); 
*trak (itota0 et etota0 et btota0); 
ttota0 = (etota0 diff itota0) et btota0; 
* 
*Verifying the enveloppe 
*ttota0 = ttota0 plus o0; 
*list (nbno ttota0); 
*elim eps0 (ttota0 et tota0); 
*Trying to fill the enveloppe 
*opti elem tet4; 
*ttota1 = (ttota0 elem tri3) et (ttota0 elem qua4 chan tri3); 
*tota1 = volu ttota1; 
*and verifying the result 
*ttota2 = enve tota1; 
*dif12 = diff ttota1 ttota2; 
*list dif12; 
* 
*Ground 
cbase0 = cont (btota0 orie direction z0); 
igrou0 = cbase0 elem appuye strictement itota0; 
egrou0 = cbase0 elem appuye strictement etota0; 
xma = maxi (coor 1 egrou0) + l0; 
xmi = mini (coor 1 egrou0) - l0; 
yma = maxi (coor 2 egrou0) + l0; 
ymi = mini (coor 2 egrou0) - l0; 
zma = maxi (coor 3 etota0) + l0; 
opti elem seg2; 
cexte0 = (xma yma 0) droi dini l0 dfin l0 
         (xma ymi 0) droi dini l0 dfin l0 
         (xmi ymi 0) droi dini l0 dfin l0 
         (xmi yma 0) droi dini l0 dfin l0; 
opti elem tri3; 
igrou0 = igrou0 surf plane coul vert; 
egrou0 = (egrou0 et cexte0) surf plane coul vert; 
eexte0 = (cexte0 tran (0 0 zma) dini l0 dfin l0) et 
         (cexte0 plus (0 0 zma) surf plane) coul blan;         
* 
*Internal space 
opti elem tet4; 
si (itota0 elem type dime neg 1); 
  itota0 = (itota0 elem tri3) et (itota0 elem qua4 chan tri3); 
fins; 
iiair0 = itota0 et igrou0; 
elim eps0 iiair0; 
*trak iiair0; 
iair0 = iiair0 volu; 
* 
*External space 
si (etota0 elem type dime neg 1); 
  etota0 = (etota0 elem tri3) et (etota0 elem qua4 chan tri3); 
fins; 
eeair0 = etota0 et egrou0 et eexte0; 
elim eps0 eeair0; 
*trak eeair0; 
eair0 = eeair0 volu; 
* 
*Total space 
tair0 = iair0 et eair0; 
elim eps0 tair0; 
*trak (enve tair0); 
* 
*Explosive bubble (center + radius) 
ec0 = 2. 2. 1.; 
er0 = 0.75; 
* 
*A partir d'un octaedre 
*ep1 = (x0*er0) plus ec0; 
*ep2 = (y0*er0) plus ec0; 
*ep3 = (z0*er0) plus ec0; 
*avec maillage sur une sphere 
*sph0 = ep1 cerc ec0 ep2 dini l0 dfin l0 
*           cerc ec0 ep3 dini l0 dfin l0 

*           cerc ec0     dini l0 dfin l0; 
*sph0 = sph0 surf spherique ec0; 
*avec maillage sur un plan 
*opti elem qua4; 
*sph0 = ep1 droi ep2 dini l0 dfin l0 
*           droi ep3 dini l0 dfin l0 
*           droi     dini l0 dfin l0; 
*l1 = mesu (elem 1 sph0); 
*sph0 = sph0 (manu chpo sph0 1 scal l1) surf plane; 
*sph0 = sph0 et (sph0 tour  90. ec0 ep3); 
*sph0 = sph0 et (sph0 tour 180. ec0 ep3); 
*sph0 = sph0 et (sph0 tour 180. ec0 ep1); 
*elim eps0 sph0; 
*sph0 = sph0 proj coni ec0 sphe ec0 ep1; 
*opti elem cub8; 
*sph0 = sph0 volu coul roug; 
* 
*A partir d'un icosaedre maille sur un plan (intersphere) 
or0 = 5.**.5+1./2.; 
ad0 = atg (-2./3.) (5.**.5/3.) + 180.; 
ep1 = (x0*er0/2.) plus (y0*er0/or0/2.) plus ec0; 
ep2 = (x0*er0/2.) moin (y0*er0/or0/2.) plus ec0; 
ep3 = (z0*er0/2.) plus (x0*er0/or0/2.) plus ec0; 
l1 = l0/2.; 
sph0 = ep1 droi ep2 dini l1 dfin l1 
           droi ep3 dini l1 dfin l1 
           droi     dini l1 dfin l1; 
l1 = mesu (elem 1 sph0); 
*opti elem qua4; 
opti elem tri3; 
sph0 = sph0 (manu chpo sph0 1 scal l1) surf plane; 
sph1 ep11 = sph0 (ep1 et ep1) tour ad0 ep2 ep3; 
ep11 = ep11 poin 1; 
sph2 ep22 = sph0 (ep2 et ep2) tour ad0 ep3 ep1; 
ep22 = ep22 poin 1; 
sph3 ep33 = sph0 (ep3 et ep3) tour ad0 ep1 ep2; 
ep33 = ep33 poin 1; 
sph0 = sph0 et sph1 et sph2 et sph3 et 
      (sph1 tour ad0 ep2 ep11) et (sph1 tour ad0 ep11 ep3) et 
      (sph2 tour ad0 ep3 ep22) et (sph2 tour ad0 ep22 ep1) et 
      (sph3 tour ad0 ep1 ep33) et (sph3 tour ad0 ep33 ep2); 
sph0 = sph0 et (sph0 syme poin ec0); 
elim eps0 sph0; 
sph0 = sph0 proj coni ec0 sphe ec0 (ep1 plus ep2 / 2.); 
*opti elem cub8; 
opti elem tet4; 
sph0 = sph0 volu coul rose; 
esp0 = tair0 incl sph0 volu larg coul roug; 
*trak esp0; 
tair0 = tair0 diff esp0 coul turq; 
*trak tair0 nosm; 
* 
stru = tota0 plus o0; 
stru6 = stru elem pri6; 
stru4 = stru elem tet4; 
aria = tair0; 

bull = esp0; 
flui = aria et bull; 
abso = eexte0; 
orie abso poin o0; 
abso = inve abso; 
* 
list (nbel bull); 
list (nbno bull); 
list (mesu bull volu); 
* 
babul = bary bull; 
pdet = bull poin proc babul; 
list babul; 
list pdet; 
* 
opti trac psc ftra 'koos21_mesh.ps'; 
trac qual bull; 
trac cach bull; 
*b1 = bull elem 1; 
*b2 = diff bull b1; 
*b1 = b1 coul roug; 
*b2 = b2 coul vert; 
*bb = b1 et b2; 
*list (mesu b1 volu); 
*list (mesu b2 volu); 
*list (mesu bb volu); 
*trac qual bb; 
*trac cach bb; 
*trac cach face bb; 
* 
blo123 = stru poin plan o0 x0 y0 eps0; 
list (nbno blo123); 
fsrn = flui poin plan o0 x0 y0 eps0; 
list (nbno fsrn ); 
fsan = enve flui chan poi1; 
list (nbno fsan); 
pabso = abso chan poi1; 
list (nbno pabso); 
pabsoxy = pabso poin plan o0 x0 y0 eps0; 
list (nbno pabsoxy); 
pabso = diff pabso pabsoxy; 
list (nbno pabso); 
fsan = diff fsan (pabso et fsrn); 
list (nbno fsan); 
* 
mesh = stru et flui et abso et blo123 et fsrn et fsan; 
tass mesh; 
opti sauv form 'koos21.msh'; 
sauv form mesh; 
list (nbel bull); 
list (mesu bull); 
list (nbel mesh); 
list (nbno mesh); 
opti donn 5; 
fin; 
 

 

The EUROPLEXUS input file reads: 
 
KOOS - 21 
$ 
 ECHO 
$VERI 
!CONV win 
CAST MESH 
TRID NONL ALE 
$ 
DIME 
 PT3L 22605 FL34 78358 TETR 6894 PRIS 5378 CL3I 5214 ZONE 4 
 NALE 1386 NBLE 8519 
TERM 
$ 
GEOM 
  FL34 flui TETR stru4 PRIS stru6 CL3I abso 
TERM 
$ 
GRIL LAGR LECT stru TERM 
     EULE LECT fsan fsrn abso TERM 
     ALE  LECT flui TERM 
     AUTO autr 
$ 
MATE 
**   l'air :  on calcule eint pour avoir P=1 bar (P=omeg*ro*eint)  
    flut ro 1.3 eint 0.21978e6 gamm 1.35 PB 0 
         ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
         CQ 2.56 PMIN 0 PREF 1.e5 NUM 11 
         a 3.738e11 b 3.747e9 r1 4.15 r2 0.90  
         ros 1630 
         LECT aria TERM 
**  Le TNT :   on donne directement ro = ros 
*              avec ignition au point PDET a l'instant 0.0  
*              la vitesse de detonation est celle de Chapman-Jouguet 
    flut ro 1630 eint 3.68e6 gamm 1.35 PB 0 

         ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
         CQ 2.56 PMIN 0 PREF 1.e5 NUM 11 
         a 3.738e11 b 3.747e9 r1 4.15 r2 0.90  
         d 6930 TDET 0.0 pini 1e5 
         xdet 1.9971 ydet 2.2741 zdet 1.0700 
         LECT bull TERM 
$ structure 
     LINE RO 2000 YOUN 3.E10 NU 0.2 
          LECT stru TERM 
$ absorbing 
     IMPE ABSI 
          LECT abso TERM 
$ 
LINK COUP 
     BLOQ 123 LECT blo123 TERM 
     FSA      LECT fsan TERM 
     FSR      LECT fsrn TERM 
$ 
CHAR CONS GRAV 0 0 -9.80665D0 LECT tous TERM 
$ 
ECRI VITE ECRO TFRE 10.E-3 
     FICH K200 TFRE 2.0E-3 POIN TOUS 
               VARI VITE ECRO ECRC LECT 1 2 TERM 
     fich spli alic      tfre 1.e-3 
$ 
OPTI NOTE 
 QUAS STAT 100. 1. UPTO 0. 
 csta 0.8e0 
 log 1 
 NF34 MOMT 2 
CALCUL TINI -20.E-3 TEND 50.E-3 
*================================================================= 
FIN 

 
Some (qualitative) results are shown. The fluid flow out of the main door: 
 

 
 



 6

Location of the explosive charge and structural deformations/velocities: 
 

 
Some selected snapshots: 
 

t = 2 ms T = 8 ms 

t = 34 ms T = 41 ms 
 


